Classroom Tips and Techniques: Teaching Fourier Series with Maple - Part 3

Robert J. Lopez

Emeritus Professor of Mathematics and Maple Fellow

Maplesoft

Initializations

```
> restart;
```

Once the files for the *FourierSeries* package by Wilhelm Werner are in place, the following command will load its code.

```
> with(FourierSeries):
```

Introduction

In this third article in a series devoted to Maple implementations of Fourier series calculations, we describe the *FourierSeries* package provided to the Maple Application Center by Wilhelm Werner. In the first article of this series, we detailed how to implement the calculations for Fourier series using just commands built into Maple. In the second, we detailed the Fourier series package by Amir Khanshan.

The *FourierSeries* Package by Wilhelm Werner

Available from the Maple Application Center, the *FourierSeries* package by Wilhelm Werner provides the five commands in Table 1. We will show how these commands meet the syntactic challenges posed by the computation and graphing of Fourier series.

Command	Input	Output
FourierCoeff	function data, series type	coefficients as function of index *n*

FourierExtend	function data, type of extension, direction of extension	odd or even extension as a function
FourierPlot	function data, series type, index value, plot interval	graph of function, and partial sum of series
FourierAnimate	function data, series type, index value, plot interval	animation of partial sums converging to function
FourierSpectrum	function data, index value	graph of $\lvert c_n \rvert$ vs. index n

Table 1 Commands in the *FourierSeries* package by Wilhelm Werner

In addition, the global input parameter listed in Table 2 determines the nature of the output of the **FourierCoeff** command.

Parameter	Effect
_FourierSeriesShowFunction	control display of the function whose Fourier coefficients are being computed, including display of piecewise linear function created by **FourierCoeff** whose input is a list of points

Table 2 Effect of global input parameter on **FourierCoeff** command

The *FourierSeries* package, and instructions for installing it, can be found on the Maple Application Center. The code is too bulky to include in this worksheet, so experiments with the commands in the package require downloading and installing the package. We have suppressed the residue of our installation process, showing only the use of the commands themselves.

FourierCoeff

Parameters	Coefficients Returned
FourierReal	a_n and b_n for trigonometric sine-cosine series, displayed
FourierComplex (FourierExp)	c_n for exponential series, as function of index
FourierCos	a_n from the trigonometric sine-cosine series, as function of index
FourierSin	b_n from the trigonometric sine-

	cosine series, as function of index

Table 3 Parameters accepted by the **FourierCoeff** command

Although it takes the four parameters listed in Table 3, the **FourierCoeff** command always computes the coefficients of the complex series. The parameter FourierReal causes **FourierCoeff** to display the coefficients of the trigonometric (sine-cosine) series, whereas the parameter FourierExp causes it to return, as a function of the index, the coefficients of the exponential form of the series.

The FourierCos and FourierSin options do not instruct the **FourierCoeff** command to create the cosine or sine series. Instead, they generate, as functions of the index n, a_n and b_n, the coefficients of the cosine terms, and sine terms, respectively, of the sine-cosine series.

To display the Fourier sine-cosine coefficients for

> $f := x \rightarrow piecewise\left(x < 0, x^2, 0 \le x, \sin(3\,x)\right)$:
> $'f'(x) = f(x)$

$$f(x) = \begin{cases} x^2 & x < 0 \\ \sin(3\,x) & 0 \le x \end{cases}$$

on $-\pi \le x \le \pi$, use the syntax

> $FourierCoeff\left([f, -\pi..\pi], \text{"FourierReal"}\right)$:

$$a_{n\sim} = \begin{cases} \frac{1}{3}\,\frac{2+\pi^3}{\pi} & n\sim = 0 \\ -\frac{2}{9} & n\sim = 3 \\ \frac{2\left(\pi\,n\sim^2 - 9\,\pi - 3\,n\sim^2\right)}{n\sim^2\left(n\sim^2 - 9\right)\pi} & n\sim = even \\ -\frac{2}{n\sim^2} & n\sim = odd \end{cases},\ b_{n\sim} =$$

$$\begin{cases} 0 & n\sim = 0 \\ -\frac{1}{54}\,\frac{-27\,\pi + 18\,\pi^2 - 8}{\pi} & n\sim = 3 \\ \frac{\pi}{n\sim} & n\sim = even \\ -\frac{\pi^2\,n\sim^2 - 4}{\pi\,n\sim^3} & n\sim = odd \end{cases}$$

Note that the quotes on the parameter FourierReal are necessary.

The piecewise functions displayed as a_n and b_n are not assigned to these names. For example, trying to obtain a_2 with the syntax

> $a(2)$

$$a(2)$$

shows no assignments have been made.

To obtain the coefficients as functions, separately generate a_n and b_n with

> $a := FourierCoeff([f, -\pi..\pi], \text{"FourierCos"}):$
> $b := FourierCoeff([f, -\pi..\pi], \text{"FourierSin"}):$

$$a_{n\sim} = \begin{cases} \frac{1}{3}\,\frac{2+\pi^3}{\pi} & n\sim = 0 \\ -\frac{2}{9} & n\sim = 3 \\ \frac{2\left(\pi\,n\sim^2 - 9\,\pi - 3\,n\sim^2\right)}{n\sim^2\left(n\sim^2 - 9\right)\pi} & n\sim = even \\ -\frac{2}{n\sim^2} & n\sim = odd \end{cases}$$

$$b_{n\sim} = \begin{cases} 0 & n\sim = 0 \\ -\frac{1}{54}\,\frac{-27\,\pi + 18\,\pi^2 - 8}{\pi} & n\sim = 3 \\ \frac{\pi}{n\sim} & n\sim = even \\ -\frac{\pi^2\,n\sim^2 - 4}{\pi\,n\sim^3} & n\sim = odd \end{cases}$$

The coefficients a_n and b_n are now functions of the index n, as we see from

> $seq(a(n), n = 0..5)$
> $seq(b(n), n = 1..5)$

$$\frac{1}{3}\frac{2+\pi^3}{\pi}, -2, \frac{1}{10}\frac{12+5\pi}{\pi}, -\frac{2}{9}, \frac{1}{56}\frac{-48+7\pi}{\pi}, -\frac{2}{25}$$

$$-\frac{-4+\pi^2}{\pi}, \frac{1}{2}\pi, -\frac{1}{54}\frac{-27\pi+18\pi^2-8}{\pi}, \frac{1}{4}\pi, -\frac{1}{125}\frac{-4+25\pi^2}{\pi}$$

Although the calls

> $a(n)$
> $b(n)$

$$\frac{2(-1)^n \pi n^2 + 18(-1)^{1+n}\pi - 3n^2 + 3(-1)^{1+n} n^2}{n^2(n^2-9)\pi}$$

$$\frac{2(-1)^{1+n} n^2 + 18(-1)^n + (-1)^n \pi^2 n^4 + 9(-1)^{1+n}\pi^2 n^2 + 2n^2 - 18}{n^3(n^2-9)\pi}$$

produce the generic expressions generated by the **int** command, a partial sum of the associated Fourier series can be obtained with the syntax

> $\frac{a(0)}{2} + add(a(n)\cos(n x) + b(n)\sin(n x), n = 1..4)$

$$\frac{1}{6}\frac{2+\pi^3}{\pi} - 2\cos(x) - \frac{(-4+\pi^2)\sin(x)}{\pi} + \frac{1}{10}\frac{(12+5\pi)\cos(2x)}{\pi} + \frac{1}{2}\pi\sin(2x) - \frac{2}{9}\cos(3x) - \frac{1}{54}\frac{(-27\pi+18\pi^2-8)\sin(3x)}{\pi} + \frac{1}{56}\frac{(-48+7\pi)\cos(4x)}{\pi} + \frac{1}{4}\pi\sin(4x)$$

Notice from the generic expressions for a_n and b_n that each expression is undefined for $n = 3$. The **add** command in Maple evaluates $a(n)$ and $b(n)$ as per their definitions as piecewise functions of n, and produces the appropriate partial sum. For $a(n)$ and $b(n)$ as created by the **FourierCoeff** command, the **sum** command (and its 2D counterpart from the Expression palette) behaves differently, and will not provide evaluation of either $a(3)$ or $b(3)$, as we see from

> $\frac{a(0)}{2} + \sum_{n=1}^{4} a(n)\cos(n x) + b(n)\sin(n x)$

```
Error, (in SumTools:-DefiniteSum:-ClosedForm) summand is singular
in the interval of summation
```

In the special case where $f(x)$ is a piecewise linear function, it can be entered as a list of points. For example, the Fourier series coefficients for the piecewise linear function

> $f := piecewise(x < 1, 2 + x, 6 - 3\,x)$

$$f := \begin{cases} 2 + x & x < 1 \\ 6 - 3\,x & otherwise \end{cases}$$

defined on $-2 \le x \le 2$, and whose graph appears in Figure 1,

> $plot(\,f, x = -2\,..2, scaling = constrained)$



Figure 1 Graph of piecewise linear function

are given by

> $FourierCoeff\,([[[-2, 0], [1, 3], [2, 0]]], "FourierReal"):$

$$a_{n\sim} = \begin{cases} 3 & n\sim = 0 \\ \dfrac{8\left(-1+\cos\left(\frac{1}{2}\,\pi\, n\sim\right)\right)}{\pi^2\, n\sim^2} & n\sim = even \\ -\dfrac{8\left(-1+\cos\left(\frac{1}{2}\,\pi\, n\sim\right)\right)}{\pi^2\, n\sim^2} & n\sim = odd \end{cases},\; b_{n\sim} = \begin{cases} 0 & n\sim = 0 \\ -\dfrac{8\sin\left(\frac{1}{2}\,\pi\, n\sim\right)}{\pi^2\, n\sim^2} & n\sim = even \\ \dfrac{8\sin\left(\frac{1}{2}\,\pi\, n\sim\right)}{\pi^2\, n\sim^2} & n\sim = odd \end{cases}$$

With a setting of the global parameter

> _FourierSeriesShowFunction := true

$$\_FourierSeriesShowFunction := true$$

the **FourierSeries** command will now provide the piecewise representation of the list of points used to describe the piecewise linear function.

> FourierCoeff([[[-2, 0], [1, 3], [2, 0]]], "FourierReal") :

$$f(t) = \begin{cases} 2+t & t < 1 \\ -3\,t+6 & t < 2 \\ -3\,t+6 & otherwise \end{cases}$$

$$a_{n\sim} = \begin{cases} 3 & n\sim = 0 \\ \dfrac{8\left(-1+\cos\left(\frac{1}{2}\, n\sim\,\pi\right)\right)}{\pi^2\, n\sim^2} & n\sim = even \\ -\dfrac{8\left(-1+\cos\left(\frac{1}{2}\, n\sim\,\pi\right)\right)}{\pi^2\, n\sim^2} & n\sim = odd \end{cases},\; b_{n\sim} = \begin{cases} 0 & n\sim = 0 \\ -\dfrac{8\sin\left(\frac{1}{2}\, n\sim\,\pi\right)}{\pi^2\, n\sim^2} & n\sim = even \\ \dfrac{8\sin\left(\frac{1}{2}\, n\sim\,\pi\right)}{\pi^2\, n\sim^2} & n\sim = odd \end{cases}$$

Unfortunately, there are no provisions for declaring the name of the independent variable, or for specifying a domain for the resulting $f(t)$. The echo of the function $f(t)$ is suppressed with

> _FourierSeriesShowFunction := false

$$\_FourierSeriesShowFunction := false$$

The coefficients c_n for the complex (or exponential) form of the Fourier series for $f(x) = x, -1 \le x \le 1$, are obtained with the syntax

> c := FourierCoeff([x→x, -1..1], "FourierExp") :

$$c_{n\sim} = \begin{cases} 0 & n\sim = 0 \\ \dfrac{I}{\pi\, n\sim} & n\sim = even \\ -\dfrac{I}{\pi\, n\sim} & n\sim = odd \end{cases}$$

A partial sum of the associated (exponential) Fourier series is

> $F := add\left(c(n)\, e^{n\pi I x}, n = -3..3\right)$

$$F := \frac{\frac{1}{3} I e^{-3 I \pi x}}{\pi} - \frac{\frac{1}{2} I e^{-2 I \pi x}}{\pi} + \frac{I e^{-I \pi x}}{\pi} - \frac{I e^{I \pi x}}{\pi} + \frac{\frac{1}{2} I e^{2 I \pi x}}{\pi} - \frac{\frac{1}{3} I e^{3 I \pi x}}{\pi}$$

The trigonometric form is then

> $evalc(F)$

$$\frac{2}{3} \frac{\sin(3\pi x)}{\pi} - \frac{\sin(2\pi x)}{\pi} + \frac{2\sin(\pi x)}{\pi}$$

Since $f(x)$ is an odd function, this is easily checked by computing

> $b := FourierCoeff([x \rightarrow x, -1..1], \text{"FourierSin"}) :$

$$b_{n\sim} = \begin{cases} 0 & n\sim = 0 \\ -\dfrac{2}{\pi\, n\sim} & n\sim = even \\ \dfrac{2}{\pi\, n\sim} & n\sim = odd \end{cases}$$

and hence,

> $add\left(b(n) \sin\left(n\pi x\right), n = 1..3\right)$

$$\frac{2}{3} \frac{\sin(3\pi x)}{\pi} - \frac{\sin(2\pi x)}{\pi} + \frac{2\sin(\pi x)}{\pi}$$

With sufficient care, the **FourierCoeff** command can also be used to display the coefficients of a sine or cosine series. For example, to obtain the sine series of the function

> $f := x \rightarrow x^2 :$
> $'f'(x) = f(x)$

$$f(x) = x^2$$

whose domain is the interval $0 \le x \le 1$, write the odd extension of $f(x)$ as

```
> g := FourierExtend( f, "odd", left) :
  g(x)
```

$$\begin{cases} -x^2 & x \le 0 \\ x^2 & 0 < x \end{cases}$$

The alternative to the **FourierExtend** command would be

```
> G := x→piecewise(x ≤ 0, -f(-x), 0 < x, f(x)) :
  'G'(x) = G(x)
```

$$G(x) = \begin{cases} -x^2 & x \le 0 \\ x^2 & 0 < x \end{cases}$$

requiring Maple coding at a more fundamental level.

Figure 2 verifies that $g(x)$ is indeed the odd extension of $f(x)$ to the interval $-1 \le x \le 1$.

```
> plot(g(x), x = -1 ..1)
```



Figure 2 The odd extension of x^2, $0 \le x \le 1$, to $[-1, 1]$

The Fourier sine-cosine series will have $a_n = 0$ and the b_n coefficients will be the coefficients of the sine series. We can see this from

> $FourierCoeff([g, -1..1], \text{"FourierReal"}):$

$$a_{n\sim} = \begin{cases} 0 & n\sim = 0 \\ 0 & n\sim = even \\ 0 & n\sim = odd \end{cases}, \quad b_{n\sim} = \begin{cases} 0 & n\sim = 0 \\ -\dfrac{2}{\pi\, n\sim} & n\sim = even \\ \dfrac{2\left(-4+\pi^2 n\sim^2\right)}{\pi^3 n\sim^3} & n\sim = odd \end{cases}$$

but the actual calculation we want to make is

> $b := FourierCoeff([g, -1..1], \text{"FourierSin"}):$

$$b_{n\sim} = \begin{cases} 0 & n\sim = 0 \\ -\dfrac{2}{\pi\, n\sim} & n\sim = even \\ \dfrac{2\left(-4+\pi^2 n\sim^2\right)}{\pi^3 n\sim^3} & n\sim = odd \end{cases}$$

By way of comparison, we can also obtain the b_n by evaluating the integral

> $qb := 2\int_0^1 f(x)\sin(n\,\pi\,x)\,dx$

$$qb := 2\left(\int_0^1 x^2 \sin(n\,\pi\,x)\,\mathrm{d}x\right)$$

whose value is

> $B := simplify(value(qb))$ assuming $n :: integer$

$$B := \frac{2\left(-2+2\,(-1)^n+(-1)^{1+n} n^2 \pi^2\right)}{n^3 \pi^3}$$

which compares favorably with

> $b(n)$

$$\frac{2\left(-2+2\,(-1)^n+(-1)^{1+n} n^2 \pi^2\right)}{n^3 \pi^3}$$

The comparison can also be make by partitioning B_n via

> $simplify(B)$ assuming $n :: even$;
> $simplify(B)$ assuming $n :: odd$

$$-\frac{2}{n\,\pi}$$

$$\frac{2\left(-4+n^2\pi^2\right)}{n^3\pi^3}$$

FourierPlot, FourierAnimate, and FourierSpectrum

The remaining three commands in the *FourierSeries* package generate graphs. When working outside the package, the standard approach is to generate the Fourier coefficients, then form a partial sum of the series, and finally, to draw an appropriate graph. Using the **FourierPlot** command, however, the graph of a partial sum is superimposed on a graph of the function without the user having to obtain either the coefficients or the partial sum. Thus, for the function

> $f := x \rightarrow piecewise(x < 0, 0, 0 \le x, x)$:
'f'(x) = f(x)

$$f(x) = \begin{cases} 0 & x < 0 \\ x & 0 \le x \end{cases}$$

we obtain Figure 3.

> $FourierPlot([f, -1..1], 3, -3..3, color = [black, red], scaling = constrained, thickness = [2, 1], discont = true)$



Figure 3 Partial sum of Fourier sine-cosine series for $f(x) = \begin{cases} 0 & x < 0 \\ x & 0 \le x \end{cases}$, $0 \le x \le 1$, superimposed on the periodic extension of the function

The partial sum graphed in Figure 3 can be obtained by writing

> $a := FourierCoeff([f, -1..1], \text{"FourierCos"}):$
$b := FourierCoeff([f, -1..1], \text{"FourierSin"}):$

$$a_{n\sim} = \begin{cases} \frac{1}{2} & n\sim = 0 \\ 0 & n\sim = even \\ -\frac{2}{\pi^2 n\sim^2} & n\sim = odd \end{cases}$$

$$b_{n\sim} = \begin{cases} 0 & n\sim = 0 \\ -\frac{1}{\pi n\sim} & n\sim = even \\ \frac{1}{\pi n\sim} & n\sim = odd \end{cases}$$

and then

> $f3 := \frac{a(0)}{2} + add(a(n)\cos(n\pi x) + b(n)\sin(n\pi x), n = 1..3)$

$$f3 := \frac{1}{4} - \frac{2\cos(\pi x)}{\pi^2} + \frac{\sin(\pi x)}{\pi} - \frac{1}{2}\frac{\sin(2\pi x)}{\pi} - \frac{2}{9}\frac{\cos(3\pi x)}{\pi^2} + \frac{1}{3}\frac{\sin(3\pi x)}{\pi}$$

Figure 4 provides proof that the parameter $k = 3$ used in the **FourierPlot** command gives rise to the partial sum

$$f_k = \frac{a_0}{2} + \sum_{n=1}^{k} \left(a_n \cos\left(\frac{n\pi x}{L}\right) + b_n \sin\left(\frac{n\pi x}{L}\right) \right)$$

> $plot([f(x), f3], x = -1..1, color = [black, red], thickness = [2, 1])$



Figure 4 Graph of the partial sum f_3 and $f(x)$

Figure 5 contains the animation provided by the **FourierAnimate** command, and shows how a sequence of partial sums converges to $\tilde{f}$.

> $FourierAnimate([f, -1..1], 10, -3..3.1, color = [black, red], discont = true, thickness = [2, 1], scaling = constrained)$



Figure 5 Animation showing convergence of partial sums f_k, $k = 1, \ldots, 10$

Finally, Figure 6 contains a graph of the frequency spectrum, a bar graph representing the points $\left(\frac{n\pi}{L}, |c_n|\right)$, drawn by the **FourierSpectrum** command. (It is essential to add at least one plot option to this command.)

> $FourierSpectrum([f, -1..1], 10, axes = normal)$



Figure 6 Frequency spectrum $|c_n|$ vs $\frac{n\pi}{L} = n\pi$ for $f(x)$